



Tech Info Library

Apple Color Plotter: AppleSoft BASIC Chart Maker II (3 of 6)

Revised: 11/15/84
Security: Everyone

Apple Color Plotter: AppleSoft BASIC Chart Maker II (3 of 6)

```
=====
1276 IF CM > 47 AND CM < 58 THEN X9 = 2:X8 = VAL ( CHR$(CM)):CM = 69:
      GOSUB 6000: GOTO 1200
1299 PRINT CHR$(7);: GOTO 1220
2000 GOSUB 72:M$ = "Save a Chart Specification": GOSUB 357
2100 PV = 20:M$ = "Name to Save Chart : ": GOSUB 350
2110 PV = 20:PH = 54:SL = 13: GOSUB 101:IF A$ = CHR$(27) THEN GOSUB 73:
      RETURN
2120 NAME$ = "C." + B$
2130 ONERR GOTO 2400
2135 PRINT
2140 PRINT D$;"DELETE";NAME$
2200 ONERR GOTO 2300
2205 PRINT D$;"OPEN ";NAME$
2206 PRINT D$;"WRITE ";NAME$
2210 FOR I = 1 TO 10
2220 PRINT TXT$(I): PRINT CENTER$(I): PRINT DENT(I): PRINT SIZE(I):
      PRINT BOLD$(I): PRINT COL(I): PRINT SL$(I)
2230 NEXT
2240 PRINT D$;"CLOSE"
2250 GOSUB 73: RETURN
2300 CALL 768: POKE 216,0: ONERR GOTO 97
2310 GOSUB 73: PRINT CHR$(7);
2315 PRINT D$;"CLOSE"
2320 M$="There is no room on the disk, so you'll have to ":PV=18:GOSUB 350
2330 M$ = "delete some files or use a different disk":PV = 20: GOSUB 350
2340 M$ = "Press Any Key to Continue":PV = 22: GOSUB 350
2350 GOSUB 400
2360 ONERR GOTO 2380
2370 PRINT D$;"DELETE ";NAME$
2375 CALL - 3288
2380 GOTO 2250
2400 CALL 768: POKE 216,0
2410 CALL - 3288: GOTO 2200
3000 GOSUB 72:M$ = "Load a Chart Specification": GOSUB 357
3100 PV = 20:M$ = "Name of Chart to Load : ":PH = 10: GOSUB 352
3110 PV=20:PH=40:SL=13: GOSUB 101:IF A$ = CHR$(27) THEN GOSUB 73: RETURN
```

```
3120 NAME$ = "C." + B$
3125 PV = 21:PH = 20: GOSUB 99
3126 PV = 21:PH = 20: GOSUB 99: PRINT " "
3130 ONERR GOTO 3300
3200 PRINT D$;"OPEN ";NAME$
3201 PRINT D$;"READ ";NAME$
3202 PV = 24:PH = 80: GOSUB 99
3210 FOR I = 1 TO 10
3220 INPUT TXT$(I): INPUT CENTER$(I): INPUT DENT(I):
      INPUT SIZE(I): INPUT BOLD$(I): INPUT COL(I): INPUT SL$(I)
3230 NEXT
3236 PRINT D$;"CLOSE"
3240 GOSUB 7005
3250 RETURN
3300 CALL 768: POKE 216,0: ONERR GOTO 97
3305 GOSUB 73:M$ = "Could NOT find Chart named " + NAME$:PV = 20: GOSUB 350
3310 M$ = "Press Any Key to Continue":PV = 22: GOSUB 350:PRINT CHR$(7);
3315 CALL - 3288
3320 GOSUB 400: GOSUB 72: GOTO 3250
4000 m$ = "This Function Not Implemented, Yet":pv = 20: GOSUB 353
4010 m$ = "Press Any Key to Continue":pv = 22: GOSUB 353
4015 PRINT CHR$(7);
4020 GOSUB 400
4030 RETURN
6000 GOSUB 72:M$ = "Enter/Change a Line": GOSUB 357
6010 M$="Line (1-10) :          Size (1-3) :          Color (1-4) :":PV=18:PH=5:
      GOSUB 352
6020 M$="Center(y/n):  Bold(y/n):  Slant(y/n):  Indent Spaces:"
      PV = 20:PH = 5: GOSUB 352
6030 M$ = "Text : ":PV = 22:PH = 5: GOSUB 352
6100 PV = 18:PH = 18:SL = 2:LL = 1:UL = 10
6102 IF X8 = 0 AND X9 = 2 THEN X8 = 10
6103 IF X8 < > 0 THEN GOSUB 99: PRINT X8;: GOTO 6130
6105 GOSUB 201: IF A$ = CHR$(27) THEN GOTO 6995
6110 ON X9 GOTO 6120,6130
6120 GOSUB 369:X8 = 0: GOTO 6100
6130 N = X8
6200 PV=18:PH=39:SL=1:LL=1:UL=3:GOSUB 201:IF A$=CHR$(27) THEN GOTO 6995
6210 ON X9 GOTO 6220,6230
6220 GOSUB 369: GOTO 6200
6230 SIZE = X8
6300 PH = 58:SL = 1:LL = 1:UL = 4: GOSUB 201:IF A$ = CHR$(27) THEN GOTO 6990
6310 ON X9 GOTO 6320,6330
6320 GOSUB 369: GOTO 6300
6330 COL = X8
6400 PV = 20:PH = 18:SL = 1: GOSUB 101: IF A$ = CHR$(27) THEN GOTO 6990
6405 IF A$ = CHR$(13) AND LEN (B$) = 0 THEN CENTER$ = "N":
      M$ = CENTER$: GOSUB 352: GOTO 6500
6410 IF B$ = "Y" OR B$ = "y" THEN CENTER$ = "Y": GOTO 6500
6420 IF B$ = "N" OR B$ = "n" THEN CENTER$ = "N": GOTO 6500
6430 GOTO 6400
6500 PV = 20:PH = 39:SL = 1: GOSUB 101:IF A$ = CHR$(27) THEN GOTO 6990
6505 IF A$ = CHR$(13) AND LEN (B$) = 0 THEN BOLD$ = "N":
```

```
      M$ = BOLD$: GOSUB 352: GOTO 6550
6510 IF B$ = "Y" OR B$ = "y" THEN BOLD$ = "Y": GOTO 6550
6520 IF B$ = "N" OR B$ = "n" THEN BOLD$ = "N": GOTO 6550
6530 GOTO 6500
6550 PV = 20:PH = 58:SL = 1: GOSUB 101: IF A$ = CHR$(27) THEN GOTO 6990
6555 IF A$=CHR$(13) AND LEN (B$) = 0 THEN SL$="N":M$=SL$:GOSUB 352:GOTO 6600
```

Apple Technical Communications

Tech Info Library Article Number:334